

Praktische Informatik 3: Einführung in die Funktionale Programmierung
Vorlesung vom 12.01.2011: Aktionen und Zustände

Christoph Lüth & Dennis Walter

Universität Bremen

Wintersemester 2010/11

Rev. 1312

1 [26]

Fahrplan

- ▶ Teil I: Funktionale Programmierung im Kleinen
- ▶ Teil II: Funktionale Programmierung im Großen
 - ▶ Abstrakte Datentypen
 - ▶ Signaturen und Eigenschaften
 - ▶ Aktionen und Zustände
- ▶ Teil III: Funktionale Programmierung im richtigen Leben


2 [26]

Inhalt

- ▶ Ein/Ausgabe in funktionale Sprachen
- ▶ Wo ist das Problem?
- ▶ Aktionen und der Datentyp *IO*.
- ▶ Aktionen als Werte
- ▶ Aktionen als Zustandstransformationen

3 [26]

Ein- und Ausgabe in funktionalen Sprachen



Problem:

- ▶ Funktionen mit Seiteneffekten nicht referentiell transparent.
- ▶ `readString :: ... -> String ??`

Lösung:

- ▶ Seiteneffekte am Typ erkennbar
- ▶ Aktionen können nur mit Aktionen komponiert werden
- ▶ „einmal Aktion, immer Aktion“

4 [26]

Aktionen als abstrakter Datentyp

- ▶ ADT mit Operationen Komposition und Lifting

- ▶ Signatur:

```
type IO α
(≫=)  :: IO α -> (α -> IO β) -> IO β
return :: α -> IO α
```

- ▶ Plus elementare Operationen (lesen, schreiben etc)

5 [26]

Elementare Aktionen

- ▶ Zeile von stdin lesen:

```
getLine :: IO String
```

- ▶ Zeichenkette auf stdout ausgeben:

```
putStr  :: String -> IO ()
```

- ▶ Zeichenkette mit Zeilenvorschub ausgeben:

```
putStrLn :: String -> IO ()
```

6 [26]

Einfache Beispiele

- ▶ Echo einfach

```
echo1 :: IO ()
echo1 = getLine ≻= putStrLn
```

- ▶ Echo mehrfach

```
echo :: IO ()
echo = getLine ≻= putStrLn ≻= λ_ -> echo
```

- ▶ Was passiert hier?

- ▶ Verknüpfen von Aktionen mit `≻=`
- ▶ Jede Aktion gibt Wert zurück

7 [26]

Noch ein Beispiel

- ▶ Umgekehrtes Echo:

```
ohce :: IO ()
ohce = getLine
      ≻= λs -> putStrLn (reverse s)
      ≻= ohce
```

- ▶ Was passiert hier?

- ▶ Reine Funktion `reverse` wird innerhalb von Aktion `putStrLn` genutzt
- ▶ Folgeaktion `ohce` benötigt Wert der vorherigen Aktion nicht
- ▶ Abkürzung: `≻`

```
p ≻ q = p ≻= λ_ -> q
```

8 [26]

Die do-Notation

- ▶ Syntaktischer Zucker für IO:

```
echo =
  getLine
  >>= λs → putStrLn s
  >> echo

echo =
  do s ← getLine
     putStrLn s
     echo
```

- ▶ Rechts sind $\gg=$, \gg implizit.
- ▶ Es gilt die **Abseitsregel**.
- ▶ Einrückung der ersten Anweisung nach **do** bestimmt Abseits.

9 [26]

Drittes Beispiel

- ▶ Zählendes, endliches Echo

```
echo3 :: Int → IO ()
echo3 cnt = do
  putStr (show cnt ++ ": ")
  s ← getLine
  if s /= "" then do
    putStrLn $ show cnt ++ ": " ++ s
    echo3 (cnt+1)
  else return ()
```

- ▶ Was passiert hier?
 - ▶ Kombination aus Kontrollstrukturen und Aktionen
 - ▶ **Aktionen als Werte**
 - ▶ Geschachtelte **do**-Notation

10 [26]

Module in der Standardbibliothek

- ▶ Ein/Ausgabe, Fehlerbehandlung (Modul IO)
- ▶ Zufallszahlen (Modul Random)
- ▶ Kommandozeile, Umgebungsvariablen (Modul System)
- ▶ Zugriff auf das Dateisystem (Modul Directory)
- ▶ Zeit (Modul Time)

11 [26]

Ein/Ausgabe mit Dateien

- ▶ Im Prelude vordefiniert:

- ▶ Dateien schreiben (überschreiben, anhängen):

```
type FilePath = String
writeFile :: FilePath → String → IO ()
appendFile :: FilePath → String → IO ()
```

- ▶ Datei lesen (verzögert):

```
readFile :: FilePath → IO String
```

- ▶ Mehr Operationen im Modul IO der Standardbibliothek

- ▶ Buffered/Unbuffered, Seeking, &c.
- ▶ Operationen auf Handle

12 [26]

Beispiel: Zeichen, Wörter, Zeilen zählen (wc)

```
wc :: String → IO ()
wc file =
  do cont ← readFile file
     putStrLn $ file ++ ": " ++
       show (length (lines cont),
            length (words cont),
            length cont)
```

- ▶ Nicht sehr effizient — Datei wird im Speicher gehalten.

13 [26]

Beispiel: wc verbessert.

- ▶ Effizienter: Dateiinhalt **einmal** traversieren

```
cnt :: Int → Int → Int → Bool → String
      → (Int, Int, Int)
cnt l w c _ [] = (l, w, c)
cnt l w c bl (x:xs)
  | isSpace x && not bl = cnt l' (w+1) (c+1) True xs
  | isSpace x && bl    = cnt l' w (c+1) True xs
  | otherwise        = cnt l w (c+1) False xs
  where l' = if x == '\n' then l+1 else l
```

- ▶ Hauptprogramm:

```
wc :: String → IO ()
wc file = do
  cont ← readFile file
  putStrLn $ file ++ ": " ++ show (cnt 0 0 0 False cont)
```

- ▶ Datei wird verzögert gelesen und dabei verbraucht.

14 [26]

Aktionen als Werte

- ▶ **Aktionen** sind **Werte** wie alle anderen.
- ▶ Dadurch **Definition** von **Kontrollstrukturen** möglich.
- ▶ Endlosschleife:

```
forever :: IO α → IO α
forever a = a >> forever a
```

- ▶ Iteration (feste Anzahl):

```
forN :: Int → IO α → IO ()
forN n a | n == 0 = return ()
         | otherwise = a >> forN (n-1) a
```

- ▶ Vordefinierte Kontrollstrukturen (Control.Monad):

- ▶ when, mapM, forM, sequence, ...

15 [26]

Fehlerbehandlung

- ▶ **Fehler** werden durch **IOError** repräsentiert

- ▶ Fehlerbehandlung durch **Ausnahmen** (ähnlich Java)

```
ioError :: IOError → IO α — "throw"
catch :: IO α → (IOError → IO α) → IO α
```

- ▶ Fehlerbehandlung nur in Aktionen

16 [26]

Fehler fangen und behandeln

- ▶ Fehlerbehandlung für wc:

```
wc2 :: String → IO ()
wc2 file =
  catch (wc file)
    (\e → putStrLn $ "Fehler:␣" ++ show e)
```

- ▶ IOError kann analysiert werden (siehe Modul IO)
- ▶ read mit Ausnahme bei Fehler (statt Programmabbruch):

```
readIO :: Read a ⇒ String → IO a
```

17 [26]

So ein Zufall!

- ▶ Zufallswerte:

```
randomRIO :: (α, α) → IO α
```

- ▶ Warum ist randomIO Aktion?

- ▶ Beispiel: Aktionen zufällig oft ausführen

```
atmost :: Int → IO α → IO [α]
atmost most a =
  do l ← randomRIO (1, most)
     mapM id (replicate l a)
```

- ▶ Zufälligen String erzeugen

```
randomStr :: IO String
randomStr = atmost 40 (randomRIO ('a', 'z'))
```

18 [26]

Ausführbare Programme

- ▶ Eigenständiges Programm ist **Aktionen**
- ▶ **Hauptaktion**: main in Modul Main
- ▶ wc als eigenständiges Programm:

```
module Main where

import System.Environment (getArgs)
import Char(isSpace)

main = do
  args ← getArgs
  mapM wc2 args
```

19 [26]

Funktionen mit Zustand

Theorem (Currying)

Folgende Typen sind *isomorph*:

$$A \times B \rightarrow C \cong A \rightarrow B \rightarrow C$$

- ▶ In Haskell: folgende Funktionen sind *invers*:

```
curry :: ((α, β) → γ) → α → β → γ
uncurry :: (α → β → γ) → (α, β) → γ
```

20 [26]

Funktionen mit Zustand

- ▶ Idee: Seiteneffekt **explizit** machen
- ▶ Funktion $f : A \rightarrow B$ mit Seiteneffekt in **Zustand** S :

$$\begin{aligned} f : A \times S &\rightarrow B \times S \\ &\cong \\ f : A \rightarrow S &\rightarrow B \times S \end{aligned}$$

- ▶ Datentyp: $S \rightarrow B \times S$
- ▶ Komposition: Funktionskomposition und uncurry

21 [26]

In Haskell: Zustände **explizit**

- ▶ Datentyp: Berechnung mit Seiteneffekt in Typ Σ :

```
type State Σ α = Σ → (α, Σ)
```

- ▶ Komposition zweier solcher Berechnungen:

```
comp :: State Σ α → (α → State Σ β) → State Σ β
comp f g = uncurry g ∘ f
```

- ▶ Lifting:

```
lift :: α → State Σ α
lift = curry id
```

22 [26]

Beispiel: Ein Zähler

- ▶ Datentyp:

```
type WithCounter α = State Int α
```

- ▶ Zähler erhöhen:

```
tick :: WithCounter ()
tick i = ((), i+1)
```

- ▶ Zähler auslesen:

```
read :: WithCounter Int
read i = (i, i)
```

- ▶ Zähler zurücksetzen:

```
reset :: WithCounter ()
reset i = ((), 0)
```

23 [26]

Implizite vs. explizite Zustände

- ▶ Nachteil: Zustand ist **explizit**

- ▶ Kann dupliziert werden

- ▶ Daher: Zustand **implizit** machen

- ▶ Datentyp verkapseln

- ▶ Signatur State, comp, lift, elementare Operationen

24 [26]

Aktionen als Zustandstransformationen

- ▶ **Idee:** Aktionen sind **Transformationen** auf Systemzustand S
- ▶ S beinhaltet
 - ▶ Speicher als Abbildung $A \rightarrow V$ (Adressen A , Werte V)
 - ▶ Zustand des Dateisystems
 - ▶ Zustand des Zufallsgenerators
- ▶ In Haskell: Typ `RealWorld`
 - ▶ "Virtueller" Typ, Zugriff nur über elementare Operationen
 - ▶ Entscheidend nur **Reihenfolge** der Aktionen

25 [26]

Zusammenfassung

- ▶ Ein/Ausgabe in Haskell durch **Aktionen**
- ▶ **Aktionen** (Typ `IO α`) sind seiteneffektbehaftete Funktionen
- ▶ **Komposition** von Aktionen durch

```
(>>=) :: IO α → (α → IO β) → IO β
return :: α → IO α
```
- ▶ **do**-Notation
- ▶ Fehlerbehandlung durch Ausnahmen (`IOError`, `catch`).
- ▶ Verschiedene Funktionen der Standardbibliothek:
 - ▶ Prelude: `getLine`, `putStr`, `putStrLn`, `readFile`, `writeFile`
 - ▶ Module: `IO`, `Random`
- ▶ Aktionen sind **implementiert** als **Zustandstransformationen**

26 [26]